

# Exhibit 9

The Stata Journal (2014)  
14, Number 3, pp. 511–540

## Merger simulation with nested logit demand

Jonas Björnerstedt  
Swedish Competition Authority  
Stockholm, Sweden  
jonas@bjornerstedt.org

Frank Verboven  
University of Leuven  
Leuven, Belgium  
frank.verboven@kuleuven.be

**Abstract.** In this article, we show how to implement merger simulation in Stata as a postestimation command, that is, after estimating an aggregate nested logit demand system with a linear regression model. We also show how to implement merger simulation when the demand parameters are not estimated but instead calibrated to be consistent with outside information on average price elasticities and profit margins. We allow for a variety of extensions, including the role of (marginal) cost savings, remedies (divestiture), and conduct different from Bertrand–Nash behavior.

**Keywords:** st0349, mergersim, merger simulation, aggregate nested logit model, unit demand and constant expenditures demand

### 1 Introduction

Competition and antitrust authorities have long been concerned with the possible anticompetitive effects of mergers. This is in particular the case for horizontal mergers, which are mergers between firms selling substitute products. The traditional concern has been that such mergers raise market power, which may hurt consumers and reduce total welfare (the sum of producer and consumer surplus). At the same time, however, it has been recognized that mergers may also result in cost savings or other efficiencies. While such cost savings may often be insufficient to reduce prices and benefit consumers, it has been shown that even small cost savings can be sufficient to raise total welfare (see [Williamson \[1968\]](#) and [Farrell and Shapiro \[1990\]](#)).<sup>1</sup> Despite the possible total welfare gains, most competition authorities in practice take a consumer surplus standard when evaluating proposed mergers.

Merger simulation is increasingly used as a tool to evaluate the effects of horizontal mergers. Consistent with policy practice, the focus is often on the price and consumer surplus effects, but various applications also evaluate the effects on total welfare.<sup>2</sup> Merger simulation aims to predict the merger effects in the following three steps.

1. According to [Williamson's \(1968\)](#) analysis, the deadweight loss from the output reduction after the merger is a second-order effect that is easily compensated by the cost savings from the merger. However, [Posner \(1975\)](#) argues that there is another source of inefficiency from mergers because firms must spend wasteful resources to make a merger and maintain market power. In this alternative view, it may be more natural to use consumer surplus as a standard to evaluate mergers and to ignore the transfer from consumers to firms.
2. Early contributions to the merger simulation literature are [Werden and Froeb \(1994\)](#), [Nevo \(2000\)](#), [Epstein and Rubinfeld \(2002\)](#), and [Ivaldi and Verboven \(2005\)](#). For a recent survey, see [Budzinski and Ruhmer \(2010\)](#).

The first step specifies and estimates a demand system, usually one with differentiated products. The second step makes an assumption about the firms' equilibrium behavior, typically multiproduct Bertrand–Nash, to compute the products' current profit margins and their implied marginal costs. The third step usually assumes that marginal costs are constant and computes the postmerger price equilibrium, accounting for increased market power, cost efficiencies, and perhaps remedies (such as divestiture). This enables one to compute the merger's effect on prices, consumer surplus, producer surplus, and total welfare. Stata is often used to estimate the demand system (the first step) but not to implement a complete merger simulation (including the second and third steps). In this article, we show how to implement merger simulation in Stata as a postestimation command, that is, after estimating the parameters of a demand system for differentiated products. We also illustrate how to perform merger simulation when the demand parameters are not estimated but rather calibrated to be consistent with outside industry information on price elasticities and profit margins. We allow for a variety of extensions, including the role of (marginal) cost savings, remedies (divestiture), and conduct different from Bertrand–Nash behavior.

We consider an oligopoly model with multiproduct price-setting firms that may partially collude and have constant marginal cost. Following [Berry \(1994\)](#), we specify the demand system as an aggregate nested logit model, which can be estimated with market-level data using linear regression methods (as opposed to the individual-level nested logit model). We consider both a unit demand specification, as in [Berry \(1994\)](#) and [Verboven \(1996\)](#), and a constant expenditures specification, as in [Björnerstedt and Verboven \(2013\)](#). The model requires a dataset on products sold in one market, or in a panel of markets, with information on the products' prices, their quantities sold, firm and nest identifiers, and possibly other product characteristics.

In section 2, we discuss the merger simulation model, including the nested logit demand system. In section 3, we introduce the commands required to carry out the merger simulation. Section 4 provides examples and section 5 concludes.

## 2 Merger simulation with an aggregate nested logit demand system

### 2.1 Merger simulation

Suppose there are  $J$  products, indexed by  $j = 1, \dots, J$ . The demand for product  $j$  is  $q_j(\mathbf{p})$ , where  $\mathbf{p}$  is a  $J \times 1$  price vector, and its marginal cost is constant and equal to  $c_j$ . Each firm  $f$  owns a subset of products  $F_f$  and chooses the prices of its own products  $j \in F_f$  to maximize

$$\Pi_f(\mathbf{p}) = \sum_{j \in F_f} (p_j - c_j) q_j(\mathbf{p}) + \phi \sum_{j \notin F_f} (p_j - c_j) q_j(\mathbf{p})$$

where  $\phi \in (0, 1)$  is a conduct parameter to allow for the possibility that firms partially coordinate. If  $\phi = 0$ , firms behave noncooperatively as multiproduct firms. If  $\phi = 1$ ,

they behave as a perfect, joint-profit maximizing cartel. A Bertrand–Nash equilibrium is defined by the following system of first-order conditions:

$$q_j(\mathbf{p}) + \sum_{k \in F_j} (p_k - c_k) \frac{\partial q_k(\mathbf{p})}{\partial p_j} + \phi \sum_{k \notin F_j} (p_k - c_k) \frac{\partial q_k(\mathbf{p})}{\partial p_j} = 0, \quad j = 1, \dots, J \quad (1)$$

Let  $\boldsymbol{\theta}$  be a  $J \times J$  product-ownership matrix, with  $\theta(j, k) = 1$  if products  $j$  and  $k$  are produced by the same firm and  $\theta(j, k) = \phi$  otherwise. If  $\phi = 0$  (no collusion),  $\boldsymbol{\theta}$  becomes the usual block diagonal matrix; if all firms own only one product,  $\boldsymbol{\theta}$  becomes the identity matrix. Furthermore, let  $\mathbf{q}(\mathbf{p})$  be the  $J \times 1$  demand vector,  $\boldsymbol{\Delta}(\mathbf{p}) \equiv \partial \mathbf{q}(\mathbf{p}) / \partial \mathbf{p}'$  be the  $J \times J$  Jacobian of first derivatives, and  $\mathbf{c}$  be the  $J \times 1$  marginal cost vector. We can then write (1) in vector notation as

$$\mathbf{q}(\mathbf{p}) + \{\boldsymbol{\theta} \odot \boldsymbol{\Delta}(\mathbf{p})\} (\mathbf{p} - \mathbf{c}) = 0$$

This can be inverted to write price as the sum of marginal cost and a markup, where the markup term (inversely) depends on the price elasticities and on the product-ownership matrix:

$$\mathbf{p} = \mathbf{c} - \{\boldsymbol{\theta} \odot \boldsymbol{\Delta}(\mathbf{p})\}^{-1} \mathbf{q}(\mathbf{p}) \quad (2)$$

For single-product firms with no collusion ( $\phi = 0$ ), the markup term is price divided by the own-price elasticity of demand. With multiproduct-firms and partial collusion, the cross-price elasticities also matter, and this increases the markup term (if products are substitutes).

Equation (2) serves two purposes. First, it can be rewritten to uncover the premerger marginal cost vector  $\mathbf{c}$  based on the premerger prices and estimated price elasticities of demand; that is,

$$\mathbf{c}^{\text{pre}} = \mathbf{p}^{\text{pre}} + \{\boldsymbol{\theta}^{\text{pre}} \odot \boldsymbol{\Delta}(\mathbf{p}^{\text{pre}})\}^{-1} \mathbf{q}(\mathbf{p}^{\text{pre}})$$

Second, (2) can be used to predict the postmerger equilibrium. The merger involves two possible changes: a change in the product ownership matrix from  $\boldsymbol{\theta}^{\text{pre}}$  to  $\boldsymbol{\theta}^{\text{post}}$  and, if there are efficiencies, a change in the marginal cost vector from  $\mathbf{c}^{\text{pre}}$  to  $\mathbf{c}^{\text{post}}$ . To simulate the new price equilibrium, one may use fixed point iteration on (2), possibly with a dampening parameter in the markup term, or another algorithm such as the Newton method (see, for example, Judd [1998, 633]).

## 2.2 Nested logit demand system

The demand system  $\mathbf{q} = \mathbf{q}(\mathbf{p})$  for the  $J$  products,  $j = 1, \dots, J$ , is specified as a nested logit model with two levels of nests, referred to as groups and subgroups. This model belongs to McFadden's (1978) generalized extreme value discrete choice model. Consumers choose the alternative that maximizes random utility, which results in a specification for choice probabilities for each alternative. The nested logit model relaxes the independence of an irrelevant alternative property of the simple logit model and allows consumers to have correlated preferences for products that belong to the same subgroup or group. While discrete choice models were initially developed to analyze individual-level

data (see Train [2009] for an overview), Berry (1994) and Berry, Levinsohn, and Pakes (1995) show how to estimate the models with aggregate data. The dataset consists of  $J \times 1$  vectors of the products' quantities  $\mathbf{q}$ , prices  $\mathbf{p}$ , and a  $J \times K$  matrix of product characteristics  $\mathbf{x}$ , including indicator variables for the products' subgroup and group and their firm affiliation. The dataset is for either one market or a panel of markets, for example, different years or different regions and countries. The panel is not necessarily balanced, because new products may be introduced over time, or old products may be eliminated, and not all products may be for sale in all regions.

In addition to each product  $j$ 's quantity sold  $q_j$ , its price  $p_j$ , and the vector of product characteristics  $x_j$ , it is necessary to observe (or estimate) the potential market size for the differentiated products. In the common unit demand specification of the nested logit, consumers have inelastic conditional demands: they buy either a single unit of their most preferred product  $j = 1, \dots, J$  or the outside good  $j = 0$ . The potential market size is then the potential number of consumers  $I$ , for example, an assumed fraction  $\gamma$  of the observed population in the market,  $I = \gamma L$ . An alternative is the constant expenditures specification, where consumers have unit elastic conditional demand: they buy a constant expenditure of their preferred product or the outside good. Here the potential market size is the potential total budget  $B$ , for example, an assumed fraction  $\gamma$  of total gross domestic product in the market,  $B = \gamma Y$ .

As shown by Berry (1994) and the extensions by Verboven (1996) and Björnerstedt and Verboven (2013), the aggregate two-level nested logit model gives rise to the following linear estimating equation for a cross section of products  $j = 1, \dots, J$ :

$$\ln(s_j/s_0) = x_j\beta + \alpha\tilde{p}_j + \sigma_1 \ln(s_{j|h_g}) + \sigma_2 \ln(s_{h|g}) + \xi_j \quad (3)$$

A subscript  $t$  can be added to consider multiple markets or time periods, as in most empirical applications. The **price** variable is  $\tilde{p}_j = p_j$  in the unit demand specification, and  $\tilde{p}_j = \ln(p_j)$  in the constant expenditures specification. The variable  $s_j$  is the market share of product  $j$  in the potential market,  $s_{j|h_g}$  is the market share of product  $j$  in its subgroup  $h$  of group  $g$ , and  $s_{h|g}$  is the market share of subgroup  $h$  in group  $g$ . More precisely, as discussed in more detail in Björnerstedt and Verboven (2013), the market shares are quantity shares in the unit demand specification

$$s_j = \frac{q_j}{I}, \quad s_{j|h_g} = \frac{q_j}{\sum_{j \in H_{h_g}} q_j}, \quad s_{h|g} = \frac{\sum_{j \in H_{h_g}} q_j}{\sum_{h=1}^{H_{h_g}} \sum_{j \in H_{h_g}} q_j}$$

and they are expenditure shares in the constant expenditures specification

$$s_j = \frac{p_j q_j}{B}, \quad s_{j|h_g} = \frac{p_j q_j}{\sum_{j \in H_{h_g}} p_j q_j}, \quad s_{h|g} = \frac{\sum_{j \in H_{h_g}} p_j q_j}{\sum_{h=1}^{H_{h_g}} \sum_{j \in H_{h_g}} p_j q_j}$$

where  $H_{h_g}$  is the set (or number) of products of subgroup  $h$  of group  $g$ .

Furthermore, in (3),  $x_j$  is a vector of observed product characteristics, and  $\xi_j$  is the error term, which captures the product's quality that is unobserved to the econometrician. Equation (3) has the following parameters to be estimated: a vector of

mean valuations  $\beta$  for the observed product characteristics, a price parameter  $\alpha < 0$ , and two nesting parameters  $\sigma_1$  and  $\sigma_2$ , which measure the consumers' preference correlation for products in the same subgroup and group. The model reduces to a one-level nested logit model with only subgroups as nests if  $\sigma_2 = 0$ , to a one-level nested logit model with only groups as nests if  $\sigma_1 = \sigma_2$ , and to a simple logit model without nests if  $\sigma_1 = \sigma_2 = 0$ . The mean gross valuation for product  $j$  is defined as  $\delta_j \equiv x_j\beta + \xi_j = \ln(s_j/s_0) - \alpha\tilde{p}_j - \sigma_1 \ln(s_{j|h,g}) - \sigma_2 \ln(s_{h|g})$ , so it can be computed from the product's market share, price, and the parameters  $\alpha$ ,  $\sigma_1$ , and  $\sigma_2$ .

In sum, the aggregate nested logit model is essentially a linear regression of the products' market shares on price, product characteristics, and (sub)group shares. In the unit demand specification, price enters linearly and market shares are in volumes; in the constant expenditures specification, price enters logarithmically and market shares are in values. In both cases, the unobserved product characteristics term,  $\xi_j$ , may be correlated with price and market shares, so instrumental variables should be used. Cost shifters would qualify as instruments, but these are typically not available at the product level. [Berry, Levinsohn, and Pakes \(1995\)](#) suggest using sums of the other products' characteristics (over the firm and the entire market). For the nested logit model, [Verboven \(1996\)](#) adds sums of the other product characteristics by subgroup and group.

### 3 The mergersim command

Various `mergersim` subcommands implement merger simulation as either commands before and after a linear nested logit regression to estimate  $\alpha$ ,  $\sigma_1$ , and  $\sigma_2$  or stand-alone commands where  $\alpha$ ,  $\sigma_1$ , and  $\sigma_2$  are specified by the user. With a panel dataset, one must time set the dataset before invoking the `mergersim` commands by using `xtset id time` or `tsset id time`, where `id` is the unique product identifier within the market, and `time` is the market identifier (time and region). Time setting is not required with a dataset for one market.

#### 3.1 Syntax

```
mergersim init [if] [in], marketsize(varname)
               {quantity(varname) | price(varname) | revenue(varname)} [nests(varlist)
               unitdemand cesdemand alpha(#) sigmas(# [ #]) name(string)]

mergersim market [if] [in], firm(varname) [conduct(#) name(string)]
```

```

mergersim simulate [if] [in], firm(varname) {buyer(#)
    seller(#) | newfirm(varname)} [conduct(#) name(string) buyereff(#)
    sellereff(#) efficiencias(varname) newcosts(varname) newconduct(#)
    method(fixedpoint|newton) maxit(#) dampen(#) keepvars detail]

mergersim mre [if] [in], {buyer(#) seller(#) | newfirm(varname)}
    [name(string)]

```

## 3.2 Options

### Demand and market specification

The demand and market specification are set in `mergersim init` and `mergersim market` (and in `mergersim simulate` if `mergersim market` is not explicitly invoked by the user).

`marketsize(varname)` specifies the potential size of market (total number of potential buyers in unit demand specification, total potential budget in constant expenditures specification). `marketsize()` is required with `mergersim init`.

Any two of `price()`, `quantity()`, or `revenue()` are required.

`quantity(varname)` specifies the quantity variable.

`price(varname)` specifies the price variable.

`revenue(varname)` specifies the revenue variable.

`nests(varlist)` specifies one or two nesting variables. The outer nest is specified first.

If only one variable is specified, a one-level nested logit model applies. If the option is not specified, a simple logit model applies.

`unitdemand` specifies the unit demand specification (default).

`cesdemand` specifies the constant expenditure specification rather than the default unit demand specification.

`alpha(#)` specifies a value for the alpha parameter rather than using an estimate. Note that this option has no effect if `mergersim market` has been run.

`sigmas(# [#])` specifies a value for the sigma parameters rather than using an estimate. In the two-level nested logit, the first sigma corresponds to the log share of the product in the subgroup, and the second corresponds to the log share of the subgroup in the group.

`name(string)` specifies a name for the simulation. Variables created will have the specified name followed by an underscore character rather than the default `M_`. This option can be used with all the `mergersim` subcommands.

**firm**(*varname*) specifies the integer variable, indexing the firm owning the product.  
**firm**() is required with **mergersim market** and **mergersim simulate**.

**conduct**(#) measures the fraction of the competitors' profits that firms account for when setting their own prices. It gives the degree of joint profit maximization between firms before the merger in percentage terms (number between 0 and 1).

### Merger specification

The merger specification is set in **mergersim simulate** or in **mergersim mre**.

Either the identity of buyer and seller firms or the new ownership structure are required. The identity corresponds to the value in the variable specified with the **firm**() option.

**buyer**(#) specifies the buyer ID in the firm variable.

**seller**(#) specifies the seller ID in the firm variable.

**newfirm**(*varname*) specifies postmerger ownership in more detail than the buyer and seller options. For example, it can be used to simulate divestitures or two cumulative mergers by manually constructing a new firm ownership variable that differs from the firm variable specified with the **firm**() option.

Efficiency gains, in terms of percentage reduction in marginal costs, can be specified by either all seller and buyer products using the **buyereff**() and **sellereff**() options or product by product with the **efficiencies**() option.

**buyereff**(#) specifies the efficiency gain of all products of the buyer firm after the merger. A value of 0 indicates no efficiency gain. The default is **buyereff**(0). For example, to incorporate a 10% efficiency gain, specify the **buyereff**(0.1) option.

**sellereff**(#) specifies the efficiency gain of all products of the seller firm after the merger.

**efficiencies**(*varname*) specifies a variable for efficiency gains more generally (that is, product by product), where, for example, 0.2 is a 20% decrease in marginal costs, and 0 is no change.

**newcosts**(*varname*) specifies a variable for postmerger costs.

**newconduct**(#) specifies the degree of joint profit maximization between firms after the merger, in percentage terms. With a **conduct** value of 1, the profits of other firms are as important as own profits.

### Computation

The computation options can be set in **mergersim simulate**, where the postmerger Nash equilibrium is computed.



`method(fixedpoint|newton)` specifies the method used to find postmerger Nash equilibrium. The option can be specified as `fixedpoint` or `newton`. The default is `method(newton)`. The Newton method starts with one iteration of the `fixedpoint` method.

`maxit(#)` specifies the maximum number of iterations in the solver methods.

`dampen(#)` specifies an initial dampening factor lower than the default `dampen(1)` in the fixed-point method. If `fixedpoint` does not converge, the method automatically tries a dampening factor of half the initial dampening.

### Display and results

`keepvars` specifies that all generated variables should be kept after simulation, calculation of elasticities, or minimal required efficiencies.

`detail` shows market shares in `mergersim simulate`. These market shares are relative to total sales (excluding the outside good). Market shares are in terms of volumes for the unit demand specification and in terms of value for the constant expenditure specification. Changes in consumer and producer surplus and in the Herfindahl–Hirshman index are also displayed.

## 3.3 Description

`mergersim` performs a merger simulation with the subcommands `init`, `market`, and `simulate`. `mergersim init` must be invoked first to initialize the settings. `mergersim market` calculates the price elasticities and marginal costs. `mergersim simulate` performs a merger simulation, automatically invoking `mergersim market` if the command has not been called by the user. In addition to displaying results, `mergersim` creates various variables at each step. By default, the names of these variables begin with `M_`.

First, `mergersim init` initializes the settings for the merger simulation. It is required before estimation and before a first merger simulation. It defines the upper and lower nests; the specification (unit demand or constant expenditures demand); the price, quantity, and revenue variables (two out of three); the potential market size variable; and the firm identifier (numerical variable). It also generates the variables necessary to estimate the demand parameters (alpha and sigmas) using a linear (nested) logit regression, similar to [Berry \(1994\)](#) and the extensions of [Björnerstedt and Verboven \(2013\)](#). The names of the market share and price variables to use in the regression will depend on the demand specification and are shown in the display output of `mergersim init`. Alternatively, the demand parameters can be calibrated with the `alpha()` and `sigmas()` options rather than being estimated.

Second, `mergersim market` computes the premerger conditions—the gross valuations  $\delta_j$  and marginal costs  $c_j$  of each product  $j$ —under assumptions regarding the degree of coordination. The computations are based on the last estimates of  $\alpha$ ,  $\sigma_1$ , and  $\sigma_2$  unless they are overruled by values specified by the user in the `alpha()` and

`sigmas()` options. `mergersim market` is required after `mergersim init` and before the first `mergersim simulate`. It is not necessary to specify `mergersim market` before additional `mergersim simulates` (unless one wants to specify new premerger values of  $\delta_j$  and  $c_j$ ).

Third, `mergersim simulate` computes the postmerger prices and quantities under assumptions regarding the identity of the merged firms, their cost efficiencies, and the degree of collusion (the same as before the merger). It is possible to repeat the command multiple times after estimation.

In addition to these three main subcommands, several other subcommands can provide useful information. For example, `mergersim mre` computes the minimum required efficiencies per product for the price not to increase after the merger. It can be invoked after `mergersim init`.

## 4 Examples

### 4.1 Preparing the data

To demonstrate `mergersim`, we use the dataset on the European car market, collected by [Goldberg and Verboven \(2001\)](http://www.econ.kuleuven.be/public/ndbad83/frank/cars.htm) and maintained on their webpages.<sup>3</sup> We take a reduced version of that dataset with fewer variables and a slightly more aggregate firm definition; the dataset is called `cars1.dta`. Each observation comprises a car model, year, and country. The total number of observations is 11,483: there are 30 years (1970–1999) and 5 countries (Belgium, France, Germany, Italy, and the United Kingdom), which implies an average of 77 car models per year and country. The car market is divided into five upper nests (groups) according to the segments: subcompact, compact, intermediate, standard, and luxury. Each segment is further subdivided into lower nests (subgroups) according to the origin: domestic or foreign (for example, Fiat is domestic in Italy and foreign in the other countries). Sales are new car registrations (`qu`). Price is measured in 1,000 Euro (in 1999 purchasing power). The product characteristics are horsepower (in kilowatts), fuel efficiency (in liter/100 kilometers), width (in centimeters), and height (in centimeters). The commands below are provided in a script called `example.do`.

---

3. See <http://www.econ.kuleuven.be/public/ndbad83/frank/cars.htm>.

520

*Merger simulation*

```
. use cars1
. summarize year country co segment domestic firm qu price horsepower fuel
> width height pop ngdp
```

Variable	Obs	Mean	Std. Dev.	Min	Max
year	11483	1985.43	8.540344	1970	1999
country	11483	2.918488	1.443221	1	5
co	11483	223.0364	206.6172	1	980
segment	11483	2.559087	1.289577	1	5
domestic	11483	.1886267	.3912288	0	1
firm	11483	14.49769	8.567491	1	34
qu	11483	19911.44	37803.6	51	433694
price	11483	18.49683	8.922665	5.260726	150.3351
horsepower	11483	57.26393	23.89019	13	169.5
fuel	11483	6.728904	1.709702	3.8	18.6
width	11483	164.4574	9.567716	122	188
height	11483	140.4434	4.631175	117.5	173.5
pop	11483	4.81e+07	2.18e+07	9660000	8.21e+07
ngdp	11483	1.76e+14	4.73e+14	5.18e+10	2.13e+15

A first key preparatory task is to define the two dimensions of the panel and to time set the data (unless there is only one cross-section). The first dimension is the “product”, that is, the car model (for example, Volkswagen [vw] Golf). The second dimension is the “market”, which can be defined as the country and year (for example, France in 1995).

```
. egen yearcountry=group(year country), label
. xtset co yearcountry
    panel variable:  co (unbalanced)
    time variable:  yearcountry, 1 to 150, but with gaps
                delta:  1 unit
```

Note that the panel is unbalanced because most models are not available throughout the entire period or in all countries.

A second key preparatory task is to define the potential market size. For the car market, it is sensible to adopt a unit demand specification. We specify the potential market size as total population divided by 4, a crude proxy for the number of households. In practice, the potential market size in a given year may be lower because cars are durable and consumers who just purchased a car may not consider buying a new one immediately.

```
. generate MSIZE=pop/4
```

4.2 Performing a merger simulation

Merger simulation can now proceed in three steps.

Initializing the merger simulation settings

The first step initializes the settings for the merger simulation using the command `mergersim init`. The next example specifies a two-level nested logit model where the groups are the segments and the subgroups are domestic or foreign with the segments. This requires the option `nests(segment domestic)`. The specification is the default unit demand specification. The price, quantity, market size, and firm variables are also specified.

```
. mergersim init, nests(segment domestic) price(price) quantity(qu)
> marketsize(MSIZE) firm(firm)
```

---

```
MERGERSIM: Merger Simulation Program
Version 1.0, Revision: 218
```

---

```
Unit demand two-level nested logit
```

Depvar	Price	Group shares
M_ls	price	M_lsjh M_lshg

---

```
Variables generated: M_ls M_lsjh M_lshg
```

---

`merger init` creates market share and price variables labeled with an `M_` prefix (the default prefix). The variable `M_ls` is the dependent variable  $\ln(s_j/s_0)$ , `M_lsjh` is the log of the subgroup share  $\ln(s_{j|hg})$ , and `M_lshg` is the log of the group share  $\ln(s_{h|g})$ .

We can estimate the nested logit model with a linear regression estimator using instrumental variables to account for the endogeneity of the price and market share variables. As a simplification to illustrate the approach, we consider a fixed-effects regression without instruments.

```
. xtreg M_ls price M_lsjh M_lshg horsepower fuel width height domestic year
> country2-country5, fe
Fixed-effects (within) regression      Number of obs      =      11483
Group variable: co                     Number of groups   =       351
R-sq:  within  = 0.8948                Obs per group: min =        1
      between  = 0.7576                      avg   =       32.7
      overall  = 0.8427                      max   =       146
                                         F(13,11119)        =      7271.50
corr(u_i, Xb) = -0.0147                Prob > F           =       0.0000
```

M_ls	Coef.	Std. Err.	t	P> t	[95% Conf. Interval]	
price	-.0468375	.0013002	-36.02	0.000	-.0493861	-.0442888
M_lsjh	.9047371	.0041489	218.07	0.000	.8966045	.9128696
M_lshg	.5677968	.0085109	66.71	0.000	.551114	.5844796
horsepower	.0038279	.0005921	6.46	0.000	.0026672	.0049886
fuel	-.0270919	.004539	-5.97	0.000	-.0359892	-.0181946
width	.0103757	.0016768	6.19	0.000	.0070889	.0136625
height	.0004322	.0022161	0.20	0.845	-.0039117	.0047761
domestic	.5230743	.0124205	42.11	0.000	.4987279	.5474206
year	.0017336	.0012022	1.44	0.149	-.000623	.0040902
country2	-.6621749	.01399	-47.33	0.000	-.6895977	-.6347521
country3	-.5883123	.0147382	-39.92	0.000	-.6172017	-.5594229
country4	-.7129762	.0137524	-51.84	0.000	-.7399333	-.686019
country5	-.4155907	.016715	-24.86	0.000	-.448355	-.3828265
_cons	-8.193457	2.246407	-3.65	0.000	-12.59681	-3.790101
sigma_u	.52455749					
sigma_e	.36374004					
rho	.6752947	(fraction of variance due to u_i)				

F test that all u\_i=0:      F(350, 11119) =      22.69      Prob > F = 0.0000

The parameters that will influence the merger simulations are the price parameter  $\alpha = -0.0468$  and the nesting parameters  $\sigma_1 = 0.905$  and  $\sigma_2 = 0.568$  (the coefficients of, respectively, `M_lsjh` and `M_lshg`). These estimates satisfy the following restrictions from economic theory:  $\alpha < 0$  and  $1 > \sigma_1 \geq \sigma_2 \geq 0$ . However, it is important to stress that the fixed-effects estimator is inconsistent because price and the subgroup and group market share variables are endogenous. As discussed in [Berry \(1994\)](#), an instrumental-variable estimator is required (for example, using `ivreg` or `xtivreg` with appropriate instruments). We therefore use only the results from the fixed-effects estimator for illustration.

### Analyzing premerger market conditions

The second step in the merger simulation calculates the premerger market conditions (the products' gross valuations and their marginal costs and the price elasticities of demand) using the command `mergersim market`. In the example below, these calculations are done for only the five countries in 1998. Because no values for  $\alpha$ ,  $\sigma_1$ , and  $\sigma_2$  are specified, `mergersim market` uses the parameters in the last available Stata estimation, that is, the ones from a fixed-effects regression.

```
. mergersim market if year == 1998
```

---

```
Supply: Bertrand competition
```

```
Demand: Unit demand two-level nested logit
```

---

```
Demand estimate
```

```
xtreg M_ls price M_lsjh M_lshg horsepower fuel width height domestic year
```

```
> country2-country5, fe
```

```
Dependent variable: M_ls
```

---

```
Parameters
```

```
alpha = -0.047
```

```
sigma1 = 0.905
```

```
sigma2 = 0.568
```

---

```
Own- and Cross-Price Elasticities: unweighted market averages
```

variable	mean	sd	min	max
M_ejj	-7.488	3.761	-30.454	-1.710
M_ejk	0.766	1.276	0.003	10.908
M_ejl	0.068	0.120	0.000	0.768
M_ejm	0.001	0.002	0.000	0.011

---

```
Observations: 449
```

---

```
Pre-merger Market Conditions
```

```
Unweighted averages by firm
```

firm code	price	Marginal costs	Pre-merger Lerner
BMW	20.194	17.499	0.146
Fiat	15.277	10.553	0.372
Ford	14.557	11.923	0.207
Honda	20.094	17.941	0.128
Hyundai	12.915	10.849	0.179
Kia	10.814	8.772	0.207
Mazda	14.651	12.557	0.156
Mercedes	25.598	21.569	0.162
Mitsubishi	15.955	13.825	0.145
Nissan	15.438	13.259	0.159
GM	21.054	18.633	0.135
PSA	16.243	13.533	0.194
Renault	15.518	12.837	0.203
Suzuki	9.289	7.226	0.234
Toyota	14.560	12.430	0.172
VW	18.990	16.388	0.181
Volvo	23.167	20.912	0.099
Daewoo	13.871	11.789	0.170

---

```
Variables generated: M_costs M_delta
```

---

These results imply fairly high own-price elasticities for the products in 1998,  $-7.488$  on average. The cross-price elasticities are higher for products within the same subgroup (0.766) than for products of a different subgroup (0.068) and especially for products of a different group (0.001). The Lerner index or percentage markup over marginal cost varies from 9.9% to 37.2%, with a tendency of higher percentage markups for firms with lower-priced models (a feature of most unit demand-logit models).

### Simulating the merger effects

The third step performs the actual merger simulation using the `mergersim simulate` command. The example below considers a merger where General Motors (GM) (firm = 15) sells its operations to VW (firm = 26). Note that the merger simulations would be the same if VW sold its operations to GM. We first carry out the merger simulations for Germany in 1998, where it can be considered a domestic merger (because GM sells the Opel brands, which are produced in Germany). It is assumed that there are no marginal cost savings to the seller or the buyer and that there is no partial coordination (neither before nor after the merger).

```
. mergersim simulate if year == 1998 & country == 3, seller(15) buyer(26)
> detail
```

#### Merger Simulation

			Simulation method: Newton
			Periods/markets: 1
Firm	Buyer	Seller	Number of iterations: 6
	26	15	Max price change in last it: 4.5e-06
Marginal cost savings			

#### Prices

Unweighted averages by firm

firm code	Pre-merger	Post-merger	Relative change
BMW	17.946	18.002	0.003
Fiat	15.338	15.341	0.000
Ford	13.093	13.362	0.023
Honda	15.778	15.780	0.000
Hyundai	12.912	12.912	0.000
Kia	11.276	11.276	0.000
Mazda	14.229	14.231	0.000
Mercedes	20.114	20.155	0.003
Mitsubishi	15.832	15.834	0.000
Nissan	15.101	15.103	0.000
GM	19.921	21.054	0.076
PSA	16.397	16.399	0.000
Renault	15.292	15.295	0.000
Suzuki	9.225	9.225	0.000
Toyota	13.019	13.020	0.000
VW	17.182	17.739	0.036
Volvo	22.149	22.154	0.000
Daewoo	13.483	13.484	0.000

```
Variables generated: M_price2 M_quantity2 M_price_ch (Other M_ variables
> dropped)
```

(output omitted)

The results show prices before and after the merger (in 1,000 Euro) and the percentage price change averaged by firm. This information is provided standard, even without the `detail` option at the end. The merger simulations predict that GM will on average raise its prices by 7.6%, while VW will on average raise its prices by 3.6%. The rivals respond with only very small price increases (with the exception of Ford).<sup>4</sup>

Because the new price vector is saved, one can use Stata's graphics to plot these results. Consider the following commands:

```
. generate perc_price_ch=M_price_ch*100
(11386 missing values generated)
. graph bar (mean) perc_price_ch if country==3&year==1998,
> over(firm, sort(perc_price_ch) descending label(angle(vertical)))
> ytitle(Percentage) title(Average percentage price increase per firm)
```

This produces the following plot:




Figure 1. Average percentage price increase per firm after merger of GM and VW

4. Note that one can also specify the `detail` option to display the market shares before and after the merger and the percentage point difference. If one is interested to see more detailed results, one can use additional options under `mergersim` results. One can also use standard Stata commands, such as `table`, based on the variables `M_price` (premerger price) and `M_price2` (postmerger price).



Without the `detail` option after the `mergersim simulate` command, the output reports only the price information. The `detail` option produces additional results on the following variables (premerger, postmerger, and changes): market shares by firm, the Herfindahl index, C4 and C8 ratios (market share of 4 and 8 largest firms), and consumer and producer surplus.<sup>5</sup>

Market shares by quantity  
Unweighted averages by firm

firm code	Pre-merger	Post-merger	Difference
BMW	0.074	0.079	0.005
Fiat	0.043	0.045	0.003
Ford	0.095	0.132	0.037
Honda	0.012	0.012	0.001
Hyundai	0.006	0.006	0.000
Kia	0.003	0.003	0.000
Mazda	0.025	0.027	0.002
Mercedes	0.100	0.116	0.017
Mitsubishi	0.015	0.017	0.001
Nissan	0.025	0.027	0.002
GM	0.166	0.108	-0.058
PSA	0.034	0.037	0.003
Renault	0.051	0.054	0.003
Suzuki	0.006	0.006	0.000
Toyota	0.027	0.029	0.002
VW	0.300	0.280	-0.020
Volvo	0.012	0.013	0.001
Daewoo	0.006	0.007	0.001

	Pre-merger	Post-merger
HHS:	1501	1972
C4:	66.07	71.50
C8:	86.21	88.01

	Change
Consumer surplus:	-1,839,750
Producer surplus:	1,303,353

For example, the Herfindahl index increases from 1,501 to 1,972. Consumer surplus (in Germany) drops by 1.8 billion Euro or 586 Euro per car (because 3.1 million cars were sold in Germany in 1998). This is partly compensated by an increase in producer surplus of 1.3 billion Euro.

5. In logit and nested logit models, consumer surplus (up to a constant) is given by the well-known  $\log(\text{sum})$  expression divided by the marginal utility of income. Caution is warranted in the constant expenditure specification because marginal utility is not constant. See [Train \(2009\)](#).

### 4.3 Accounting for efficiencies, remedies, and partial collusion

It is possible to account for several specific features of the merger.

#### Efficiencies

First, one may account for the possibility that the buying or the selling firm benefits from a marginal cost saving, which may be passed on to consumer prices. The cost saving is expressed as a percentage of current marginal cost. In the command below, the options `sellereff(0.2)` and `buyereff(0.2)` mean that the seller and the buyer each have a marginal cost saving of 20% on all of their products.

```
. mergersim simulate if year == 1998 & country == 3, seller(15) buyer(26)
> sellereff(0.20) buyereff(0.20) method(fixedpoint) maxit(40) dampen(0.5)
```

#### Merger Simulation

	Buyer	Seller	Simulation method: Dampened Fixed point
Firm	26	15	Periods/markets: 1
Marginal cost savings	.2	.2	Number of iterations: 19
			Max price change in last it: .

#### Prices

Unweighted averages by firm

firm code	Pre-merger	Post-merger	Relative change
BMW	17.946	17.703	-0.011
Fiat	15.338	15.265	-0.004
Ford	13.093	13.125	0.003
Honda	15.778	15.737	-0.002
Hyundai	12.912	12.908	-0.000
Kia	11.276	11.274	-0.000
Mazda	14.229	14.212	-0.001
Mercedes	20.114	19.259	-0.031
Mitsubishi	15.832	15.810	-0.001
Nissan	15.101	14.981	-0.005
GM	19.921	18.980	-0.022
PSA	16.397	16.372	-0.002
Renault	15.292	15.261	-0.003
Suzuki	9.225	9.219	-0.001
Toyota	13.019	13.005	-0.001
VW	17.182	15.717	-0.075
Volvo	22.149	22.036	-0.005
Daewoo	13.483	13.477	-0.000

```
Variables generated: M_price2 M_quantity2 M_price_ch (Other M_ variables
> dropped)
```

There is now a predicted price decrease in Germany of  $-2.2\%$  for GM and  $-7.5\%$  for VW. This implies that the 20% cost savings are sufficiently passed to consumers. To obtain convergence, we used a fixed-point iteration with a dampening factor of 0.5 because the default Newton method did not converge. `sellereff()` and `buyereff()` assume the same percentage cost saving for all products of the seller and buyer. A

more flexible option is `efficiencies()`, which enables one to have product-specific percentage cost saving based on the variable that enters in `efficiencies()`.

Instead of simulating the prices in the postmerger equilibrium with efficiencies, one can compute the minimum required efficiency (percentage cost saving by product) for the prices to remain unchanged after the merger; see [Froeb and Werden \(1998\)](#) or [Röller, Stennek, and Verboven \(2001\)](#). This can be done with the `mergersim mre` command:

```
. mergersim mre if year == 1998 & country == 3, seller(15) buyer(26)
```

Minimum Required Efficiencies for merging firms				
variable	mean	sd	min	max
M_costs	15.247	9.504	6.233	43.649
M_costs2	13.769	9.938	5.439	43.620
M_mre	0.123	0.128	0.001	0.401
Weighted average MRE: 0.221      Observations: 19				
Variable generated: M_mre				

The generated variable `M_mre` refers to the minimum required efficiency per product owned by the merging firms and is set to a missing value for the products of the non-merging firms. According to the results, the minimum required efficiencies for the 19 products of the merging firms are on average 12.3% (unweighted) and 22.1% (weighted by sales).

### Divestiture as a remedy

Second, one may account for divestiture as a remedy to mitigate the price effects of a merger. Under such a remedy, the competition authority accepts the merger on the condition that the firms sell some of their products or brands. To simulate the effects of a merger with divestiture, one can replace the options `buyer( # )` and `seller( # )` with the option `newfirm(varname)`, which specifies a variable for the new ownership structure after the merger. To illustrate, we consider a merger between Renault (firm = 18) and PSA (firm = 16), where PSA sells the brands Peugeot and Citroën. This merger would substantially raise average prices in France: 59.8% for the Renault products and 63.1% for the PSA products (ignoring entry and substitution to other countries). To mitigate the anticompetitive effects, the competition authority may request that PSA sell one of its brands, Citroën (brand = 4), to Fiat (firm = 4). The commands below show how to simulate the effects of such a merger with divestiture after creating the appropriate variable `firm_rem` for the new ownership structure.<sup>6</sup>

6. Note that this example starts with `mergersim init` and moves to `mergersim simulate` without performing a regression to obtain the price and nesting parameters. In this case, `mergersim` continues to use the most recent results.

```

. generate firm_rem=firm
. replace firm_rem=16 if firm==18      // original merger
(890 real changes made)
. replace firm_rem=4 if brand==4       // divestiture
(583 real changes made)
. quietly mergersim init, nests(segment domestic) unit price(price)
> quantity(qu) marketsize(MSIZE) firm(firm)
. quietly mergersim simulate if year == 1998 & country == 2, seller(16)
> buyer(18)
. mergersim simulate if year == 1998 & country == 2, newfirm(firm_rem)

```

---

#### Merger Simulation

		Simulation method: Newton
		Periods/markets: 1
Ownership from:	Variable name	Number of iterations: 7
Marginal cost savings	firm_rem	Max price change in last it: 9.7e-08

---

#### Prices

Unweighted averages by firm

firm code	Pre-merger	Post-merger	Relative change
BMW	18.342	18.347	0.000
Fiat	12.688	12.749	0.006
Ford	11.995	12.001	0.001
Honda	15.742	15.744	0.000
Hyundai	9.862	9.863	0.000
Kia	7.040	7.040	0.000
Mazda	12.536	12.536	0.000
Mercedes	25.239	25.240	0.000
Mitsubishi	14.880	14.880	0.000
Nissan	12.371	12.372	0.000
GM	18.963	18.966	0.000
PSA	15.303	16.317	0.089
Renault	14.996	17.114	0.162
Suzuki	7.824	7.824	0.000
Toyota	12.638	12.638	0.000
VW	17.735	17.744	0.001
Volvo	22.641	22.642	0.000
Daewoo	13.939	13.940	0.000

---

Variables generated: M\_price2 M\_quantity2 M\_price\_ch (Other M\_ variables  
> dropped)

---

The results show that the merger with divestiture raises the average price only by 16.2% for Renault and by 8.9% for the Peugeot brand, whereas the price of Fiat (now including the Citroën brand) increases by 0.6%. The option `newfirm(varname)` can also be used for other applications, for example, to assess the impact of two consecutive mergers.

**Conduct**

Third, one may account for the possibility that firms partially coordinate, that is, take into account a fraction of the competitors' profits when setting prices. Assume, for example, that firms maintain the same degree of coordination before and after the merger: one can set the conduct parameter such that the markups are in line with outside estimates. Performing `mergersim market` before `mergersim simulate` enables one to verify whether the conduct parameter results in premerger markups in line with outside estimates. This is shown in the following example (which returns to the earlier merger between GM and VW in Germany).

```
. mergersim market if year == 1998 & country == 3, conduct(0.5)
```

---

```
Supply: Partial collusion, conduct = .5
Demand: Unit demand two-level nested logit
```

---

```
Demand estimate
xtreg M_ls price M_lsjh M_lshg horsepower fuel width height domestic year
> country2-country5, fe
Dependent variable: M_ls
```

---

```
Parameters
alpha = -0.047
sigma1 = 0.905
sigma2 = 0.568
```

---

```
Own- and Cross-Price Elasticities: unweighted market averages
```

variable	mean	sd	min	max
M_ejj	-6.907	2.876	-22.039	-3.339
M_ejk	0.781	1.141	0.007	4.920
M_ejl	0.060	0.123	0.001	0.637
M_ejm	0.001	0.002	0.000	0.011

---

```
Observations: 97
```

---

Pre-merger Market Conditions  
Unweighted averages by firm

firm code	price	Marginal costs	Pre-merger Lerner
BMW	17.946	13.079	0.290
Fiat	15.338	10.845	0.334
Ford	13.093	8.114	0.419
Honda	15.778	11.433	0.286
Hyundai	12.912	8.818	0.349
Kia	11.276	7.196	0.391
Mazda	14.229	10.012	0.315
Mercedes	20.114	13.753	0.348
Mitsubishi	15.832	11.612	0.280
Nissan	15.101	10.651	0.316
GM	19.921	14.862	0.297
PSA	16.397	12.106	0.299
Renault	15.292	10.893	0.340
Suzuki	9.225	5.084	0.461
Toyota	13.019	8.794	0.379
VW	17.182	12.104	0.352
Volvo	22.149	17.596	0.208
Daewoo	13.483	9.339	0.346

Variables generated: M\_costs M\_delta

The results show that if firms coordinate by taking into account 50% of the competitors' profits, then the Lerner index becomes almost twice as high as when there is no coordination. The predicted price effects after the merger can now be computed.

```
. mergersim simulate if year == 1998 & country == 3, seller(15) buyer(26)
> conduct(0.5)
```

#### Merger Simulation

	Buyer	Seller	Simulation method: Newton
Firm	26	15	Periods/markets: 1
Marginal cost savings			Number of iterations: 6
			Max price change in last it: 2.1e-07
Conduct:	Pre	Post	
	.5	.5	

Prices  
Unweighted averages by firm

firm code	Pre-merger	Post-merger	Relative change
BMW	17.946	18.125	0.011
Fiat	15.338	15.434	0.007
Ford	13.093	13.881	0.063
Honda	15.778	15.889	0.008
Hyundai	12.912	13.019	0.009
Kia	11.276	11.379	0.009
Mazda	14.229	14.334	0.008
Mercedes	20.114	20.427	0.025
Mitsubishi	15.832	15.956	0.008
Nissan	15.101	15.194	0.007
GM	19.921	21.171	0.084
PSA	16.397	16.503	0.007
Renault	15.292	15.395	0.008
Suzuki	9.225	9.314	0.010
Toyota	13.019	13.115	0.008
VW	17.182	17.947	0.049
Volvo	22.149	22.265	0.005
Daewoo	13.483	13.584	0.008

Variables generated: M\_price2 M\_quantity2 M\_price\_ch (Other M\_ variables  
> dropped)

Under partial coordination, the merger simulation predicts larger price increases. On one hand, there is a larger predicted price increase for the merging firms: this feature does not hold generally, because the merging firms already partially coordinate before the merger. On the other hand, there is also a larger predicted price increase for the outsider firms: this feature may hold more generally because it reflects that outsiders have more cooperative responses to price changes by the merging firms.

#### 4.4 Calibrating instead of estimating the price and nesting parameters

##### Calibration

The merger simulation results depend on the values of three parameters:  $\alpha$ ,  $\sigma_1$ , and  $\sigma_2$  (and on the price and quantity data per product). A practitioner may not want to rely too heavily on the econometric estimates of these parameters and may want to verify whether the elasticities and markups are consistent with external industry information. Here a practitioner would not estimate but “calibrate” the parameters such that they result in price elasticities and markups that are equal to external estimates. Such calibration is possible by specifying the options `alpha()` and `sigmas()` to `mergersim market`. The selected values overrule the values in memory, for example, the ones from a previous estimation. In the lines below, we specify  $\alpha = -0.035$  (closer to 0 as compared with the econometric estimate of  $\alpha = -0.047$ ), and we keep  $\sigma_1$  and  $\sigma_2$  to the previous values. Hence, we calibrate  $\alpha$  such that demand would be less elastic. The results from this calibration indeed imply lower price elasticities (on average  $-5.5$ ):

---

```
. mergersim market if year == 1998 & country == 3
```

---

```
Supply: Bertrand competition
```

```
Demand: Unit demand two-level nested logit
```

---

```
Demand calibration
```

```
Parameters
```

```
alpha = -0.035
```

```
sigma1 = 0.910
```

```
sigma2 = 0.570
```

---

```
Own- and Cross-Price Elasticities: unweighted market averages
```

variable	mean	sd	min	max
M_ejj	-5.457	2.273	-17.430	-2.640
M_ejk	0.624	0.911	0.006	3.946
M_ejl	0.045	0.093	0.000	0.480
M_ejm	0.001	0.001	0.000	0.008

---

```
Observations: 97
```

---

```
Pre-merger Market Conditions
```

```
Unweighted averages by firm
```

---

firm code	price	Marginal costs	Pre-merger Lerner
BMW	17.946	14.738	0.193
Fiat	15.338	12.297	0.229
Ford	13.093	9.765	0.287
Honda	15.778	12.921	0.189
Hyundai	12.912	10.294	0.223
Kia	11.276	8.681	0.248
Mazda	14.229	11.455	0.206
Mercedes	20.114	15.030	0.255
Mitsubishi	15.832	13.019	0.186
Nissan	15.101	12.155	0.209
GM	19.921	16.573	0.199
PSA	16.397	13.576	0.197
Renault	15.292	12.302	0.236
Suzuki	9.225	6.586	0.294
Toyota	13.019	10.280	0.246
VW	17.182	13.540	0.254
Volvo	22.149	18.974	0.144
Daewoo	13.483	10.860	0.220

---

```
Variables generated: M_costs M_delta
```

---



The next lines show what this calibration implies for merger simulation.

```
. mergersim simulate if year == 1998 & country == 3, seller(15) buyer(26)
```

---

Merger Simulation

	Buyer	Seller	Simulation method: Newton
Firm	26	15	Periods/markets: 1
Marginal cost savings			Number of iterations: 6
			Max price change in last it: 5.9e-06

---

Prices

Unweighted averages by firm

firm code	Pre-merger	Post-merger	Relative change
BMW	17.946	18.018	0.004
Fiat	15.338	15.342	0.000
Ford	13.093	13.443	0.030
Honda	15.778	15.781	0.000
Hyundai	12.912	12.912	0.000
Kia	11.276	11.276	0.000
Mazda	14.229	14.231	0.000
Mercedes	20.114	20.167	0.003
Mitsubishi	15.832	15.835	0.000
Nissan	15.101	15.103	0.000
GM	19.921	21.372	0.098
PSA	16.397	16.399	0.000
Renault	15.292	15.296	0.000
Suzuki	9.225	9.226	0.000
Toyota	13.019	13.020	0.000
VW	17.182	17.892	0.045
Volvo	22.149	22.155	0.000
Daewoo	13.483	13.484	0.000

---

Variables generated: M\_price2 M\_quantity2 M\_price\_ch (Other M\_ variables  
> dropped)

---

These results show that the predicted price increase is larger when demand is less elastic.

### Application: Bootstrapping confidence intervals

One can also use the calibration options `alpha()` and `sigmas()` to implement a parametric bootstrap for constructing confidence intervals of the computed merger effects. The following lines perform three steps. First, we take 100 draws for  $\alpha$ ,  $\sigma_1$ , and  $\sigma_2$  assuming the parameters are normally distributed. Second, we perform 100 merger simulations for each draw. Third, we save the results for the average price increase of the buying firm and the selling firm, and we compute summary statistics.

```

. quietly mergersim init, nests(segment domestic) price(price) quantity(qu)
> marketsize(MSIZE) firm(firm)
. matrix b=e(b)
. matrix V=e(V)
. matrix bsub = ( b[1,1] , b[1,2] , b[1,3] )
. matrix Vsub = ( V[1,1], V[1,2], V[1,3] \ V[2,1] , V[2,2], V[2,3] \ V[3,1],
> V[3,2], V[3,3] )
. local ndraws 100
. set seed 1
. preserve
. drawnorm alpha sigma1 sigma2, n(`ndraws`) cov(Vsub) means(bsub) clear
(obs 100)
. mkmat alpha sigma1 sigma2, matrix(params)
. restore
. matrix pr_ch = J(`ndraws`,2,0)
. forvalues i = 1 2 to `ndraws` {
2. local alpha = params[`i`,1]
3. local sigma1 = params[`i`,2]
4. local sigma2 = params[`i`,3]
5. quietly mergersim init, nests(segment domestic) price(price) quantity(qu)
> marketsize(MSIZE) firm(firm) alpha(`alpha`) sigmas(`sigma1` `sigma2`)
6. quietly mergersim simulate if year == 1998 & country == 3, seller(15)
> buyer(26)
7. sum M_price_ch if year == 1998 & country == 3&firm==15, meanonly
8. matrix pr_ch[`i`,1] = r(mean)
9. sum M_price_ch if year == 1998 & country == 3&firm==26, meanonly
10. matrix pr_ch[`i`,2] = r(mean)
11. }
. clear
. quietly svmat pr_ch , names(pr_ch)
. sum pr_ch1 pr_ch2

```

Variable	Obs	Mean	Std. Dev.	Min	Max
pr_ch1	100	.0763034	.0031552	.0667844	.0844396
pr_ch2	100	.0355618	.0015778	.0307121	.0394875

Earlier, we obtained point estimates for the percentage price increase of 7.6% for GM and 3.6% for VW (for the base scenario). The 95% confidence intervals for these price increases are [6.7–8.4]% for GM and [3.1–4.0]% for VW.

## 4.5 Constant expenditures demand

We can finally illustrate how to do merger simulation based on a constant expenditures demand instead of a unit demand specification. For cars, this may not be a realistic option, because consumers typically buy one unit or no unit rather than constant expenditures. Nevertheless, we can use the constant expenditures specification to see how functional form affects the predictions from merger simulation.

First, we need to define the potential market size.

```
. generate MSIZE1=ngdpe/5
```

This assumes the potential expenditures on cars in a country and year are 20% of total gross domestic product.

Next we calibrate (rather than estimate) the parameters to  $\alpha = -0.5$ ,  $\sigma_1 = 0.9$ , and  $\sigma_2 = 0.6$ .

```
. mergersim init, nests(segment domestic) ces price(price) quantity(qu)
> marketsize(MSIZE1) firm(firm) alpha(-0.5) sigmas(0.9 .6)
(output omitted)
```

We can verify the premerger elasticities and markups at these calibrated parameters:

```
. mergersim market if year == 1998 & country == 3
```

---

```
Supply: Bertrand competition
Demand: Constant expenditure two-level nested logit
```

---

Demand calibration

Parameters

```
alpha = -0.500
sigma1 = 0.900
sigma2 = 0.600
```

---

Own- and Cross-Price Elasticities: unweighted market averages

variable	mean	sd	min	max
M_ejj	-5.574	0.493	-5.995	-4.054
M_ejk	0.426	0.493	0.005	1.946
M_ejl	0.039	0.065	0.000	0.283
M_ejm	0.001	0.001	0.000	0.006

---

Observations: 97

---

Pre-merger Market Conditions

Unweighted averages by firm

firm code	price	Marginal costs	Pre-merger Lerner
BMW	17.946	14.375	0.194
Fiat	15.338	12.451	0.189
Ford	13.093	10.502	0.202
Honda	15.778	12.938	0.180
Hyundai	12.912	10.732	0.169
Kia	11.276	9.384	0.168
Mazda	14.229	11.684	0.177
Mercedes	20.114	14.228	0.260
Mitsubishi	15.832	12.978	0.180
Nissan	15.101	12.281	0.183
GM	19.921	15.784	0.206
PSA	16.397	13.473	0.179
Renault	15.292	12.504	0.188
Suzuki	9.225	7.661	0.170
Toyota	13.019	10.739	0.175
VW	17.182	13.395	0.221
Volvo	22.149	17.606	0.201
Daewoo	13.483	11.201	0.169

Variables generated: M\_costs M\_delta

The premerger elasticities and markups are roughly comparable with the ones of the estimated unit demand model (with less variation between firms). However, as shown below, the merger simulation results in a larger predicted price increase: +10.1% for GM and +4.4% for VW. This follows from the different functional form: the constant expenditures specification has the property of quasi-constant price elasticity, whereas the unit demand specification has the property that consumers become more price sensitive as firms raise prices. For this same reason, efficiencies in the form of marginal cost savings would also be passed more to consumers under this specification.

538

*Merger simulation*

```
. mergersim simulate if year == 1998 & country == 3, seller(15) buyer(26)
> detail
```

## Merger Simulation

	Buyer	Seller	Simulation method: Newton
Firm	26	15	Periods/markets: 1
Marginal cost savings			Number of iterations: 7
			Max price change in last it: 4.7e-09

## Prices

Unweighted averages by firm

firm code	Pre-merger	Post-merger	Relative change
BMW	17.946	18.021	0.004
Fiat	15.338	15.342	0.000
Ford	13.093	13.302	0.017
Honda	15.778	15.781	0.000
Hyundai	12.912	12.912	0.000
Kia	11.276	11.276	0.000
Mazda	14.229	14.231	0.000
Mercedes	20.114	20.155	0.003
Mitsubishi	15.832	15.835	0.000
Nissan	15.101	15.103	0.000
GM	19.921	21.581	0.101
PSA	16.397	16.399	0.000
Renault	15.292	15.295	0.000
Suzuki	9.225	9.225	0.000
Toyota	13.019	13.020	0.000
VW	17.182	17.933	0.044
Volvo	22.149	22.159	0.000
Daewoo	13.483	13.484	0.000

Variables generated: M\_price2 M\_quantity2 M\_price\_ch (Other M\_ variables  
> dropped)

(output omitted)

Because the `detail` option was added, `mergersim simulate` reports additional results. Consumer surplus now drops by −2.2 billion Euro (versus −1.8 billion Euro in the unit demand specification), and producer surplus increases by 1.1 billion Euro (versus 1.3 billion Euro before).

	Pre-merger	Post-merger
HHS:	1501	1906
C4:	66.07	70.52
C8:	86.21	87.61

	Change
Consumer surplus:	−2,190,399
Producer surplus:	1,140,647

## 5 Conclusions

This overview has shown how to apply two specifications of the two-level nested logit demand system to merger simulation. We show that merger simulation can be applied as a postestimation command based on estimated parameter values, or it can be implemented without estimation but with calibrated parameters. The merger simulation results yield intuitive predictions given the assumed demand parameters.<sup>7</sup> The set of merger simulation commands can be used to simulate the effects of horizontal mergers in a standard setting (differentiated products, multiproduct Bertrand price setting). One can also incorporate various extensions, including efficiencies in the form of cost savings, remedies through partial divestiture, and alternative behavioral assumptions (partial collusive behavior).

Other applications and extensions could be considered. For example, for the car market, it could be interesting to generalize the demand model to allow consumers to substitute between countries by introducing an upper nest for the choice of country instead of assuming such substitution is not possible. These additional substitution possibilities would limit the market power effects of mergers. Other demand models may also be considered, such as a random coefficients logit model or the almost ideal demand system.

## 6 References

- Berry, S., J. Levinsohn, and A. Pakes. 1995. Automobile prices in market equilibrium. *Econometrica* 63: 841–890.
- Berry, S. T. 1994. Estimating discrete-choice models of product differentiation. *RAND Journal of Economics* 25: 242–262.
- Björnerstedt, J., and F. Verboven. 2013. Does merger simulation work? Evidence from the Swedish analgesics market. <http://www.econ.kuleuven.be/public/ndbad83/Frank/Papers/Bjornerstedt%20&%20Verboven,%202013.pdf>.
- Budzinski, O., and I. Ruhmer. 2010. Merger simulation in competition policy: A survey. *Journal of Competition Law & Economics* 6: 277–319.
- Epstein, R. J., and D. L. Rubinfeld. 2002. Merger simulation: A simplified approach with new applications. *Antitrust Law Journal* 69: 883–919.
- Farrell, J., and C. Shapiro. 1990. Horizontal mergers: An equilibrium analysis. *American Economic Review* 80: 107–126.
- Froeb, L. M., and G. J. Werden. 1998. A robust test for consumer welfare enhancing mergers among sellers of a homogeneous product. *Economics Letters* 58: 367–369.

7. We stress, however, that the estimated parameters were based on an inconsistent fixed-effects estimator. In practice, one should use instrumental variables to estimate the parameters consistently.

- Goldberg, P. K., and F. Verboven. 2001. The evolution of price dispersion in the European car market. *Review of Economic Studies* 68: 811–848.
- Ivaldi, M., and F. Verboven. 2005. Quantifying the effects from horizontal mergers in European competition policy. *International Journal of Industrial Organization* 23: 669–691.
- Judd, K. L. 1998. *Numerical Methods in Economics*. Cambridge, MA: MIT Press.
- McFadden, D. 1978. Modelling the choice of residential location. In *Spatial Interaction Theory and Planning Models*, ed. A. Karlqvist, L. Lundqvist, F. Snickars, and J. Weibull, 75–96. Amsterdam: North-Holland.
- Nevo, A. 2000. Mergers with differentiated products: The case of the ready-to-eat cereal industry. *RAND Journal of Economics* 31: 395–421.
- Posner, R. A. 1975. The social costs of monopoly and regulation. *Journal of Political Economy* 83: 807–828.
- Röller, L.-H., J. Stennek, and F. Verboven. 2001. Efficiency gains from mergers. *European Economy* 5: 31–128.
- Train, K. E. 2009. *Discrete Choice Methods with Simulation*. 2nd ed. Cambridge: Cambridge University Press.
- Verboven, F. 1996. International price discrimination in the European car market. *RAND Journal of Economics* 27: 240–268.
- Werden, G. J., and L. Froeb. 1994. The effects of mergers in differentiated products industries: Logit demand and merger policy. *Journal of Law, Economics, and Organization* 10: 407–426.
- Williamson, O. E. 1968. Economies as an antitrust defense: The welfare tradeoffs. *American Economic Review* 58: 18–36.

#### **About the authors**

Jonas Björnerstedt is a researcher at the Swedish Competition Authority and at the University of Leuven (Belgium). His current research focuses on the empirical analysis of competition policy and industrial organization.

Frank Verboven is a professor of economics and industrial organization at the University of Leuven (Belgium) and research fellow at the Centre for Economic Policy Research. His current research focuses on the empirical analysis of industries with market power, with applications to issues in competition policy and regulation.